# Maintainer: Felix Yan <felixonmars@archlinux.org>
# Contributor: Justin Davis (juster) <jrcd83@gmail.com>

pkgname=perl-mozilla-ca
pkgver=20180117
pkgrel=2
pkgdesc="Mozilla's CA cert bundle in PEM format"
arch=('any')
license=('GPL2' 'LGPL2.1')
options=('!emptydirs')
depends=('perl')
url='https://metacpan.org/release/Mozilla-CA'
source=("http://search.cpan.org/CPAN/authors/id/A/AB/ABH/Mozilla-CA-$pkgver.tar.gz")
sha512sums=('8b4187f47638ba8d670b3f5cafc1de4040dfec1dde6a932411c79b41a8a454bcd3b66f9818c6d28135e29551ce93bc3ddb61e648c7810b30cb147ea0beaaedba')

build() (
  cd Mozilla-CA-$pkgver
  perl Makefile.PL INSTALLDIRS=vendor
  make
)

check() (
  cd Mozilla-CA-$pkgver
  make test
)

package() (
  cd Mozilla-CA-$pkgver
  make install DESTDIR="$pkgdir"
)

# vim:set ts=2 sw=2 et:
